Rules for integrands of the form $(a+b\,Sin[e+f\,x])^m\,(c+d\,Sin[e+f\,x])^n\,\left(A+B\,Sin[e+f\,x]+C\,Sin[e+f\,x]^2\right)$

0: $\int(a+b\,Sin[e+f\,x])^m\,(c+d\,Sin[e+f\,x])^n\,\left(A+B\,Sin[e+f\,x]+C\,Sin[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,A\,b^2-a\,b\,B+a^2\,C==0$

- Derivation: Algebraic simplification

  Basis: If $A\,b^2-a\,b\,B+a^2\,C==0$, then $A+B\,z+C\,z^2==\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

- Rule: If $b\,c-a\,d\neq 0\,\wedge\,A\,b^2-a\,b\,B+a^2\,C==0$, then

$$\int(a+b\,Sin[e+f\,x])^m\,(c+d\,Sin[e+f\,x])^n\,\left(A+B\,Sin[e+f\,x]+C\,Sin[e+f\,x]^2\right)\,dx\,\longrightarrow$$
$$\frac{1}{b^2}\int(a+b\,Sin[e+f\,x])^{m+1}\,(c+d\,Sin[e+f\,x])^n\,(b\,B-a\,C+b\,C\,Sin[e+f\,x])\,dx$$

- Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  1/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*(b*B-a*C+b*C*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[A*b^2-a*b*B+a^2*C,0]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C/b^2*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*(a-b*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[A*b^2+a^2*C,0]
```

1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,m<-1$

Derivation: Algebraic expansion, nondegenerate sine recurrence 1c with $c\to 1,\ d\to 0,\ A\to c,\ B\to d,\ C\to 0,\ n\to 0,\ p\to 0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2 == \frac{A\,b^2-a\,b\,B+a^2\,C}{b^2}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,m<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{A\,b^2-a\,b\,B+a^2\,C}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,dx+\frac{1}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])\,(b\,B-a\,C+b\,C\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{(b\,c-a\,d)\,\left(A\,b^2-a\,b\,B+a^2\,C\right)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{b^2\,f\,(m+1)\,\left(a^2-b^2\right)}-\frac{1}{b^2\,(m+1)\,\left(a^2-b^2\right)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\cdot$$
$$\Big(b\,(m+1)\,((b\,B-a\,C)\,(b\,c-a\,d)-A\,b\,(a\,c-b\,d))+$$
$$\left(b\,B\,\left(a^2\,d+b^2\,d\,(m+1)-a\,b\,c\,(m+2)\right)+(b\,c-a\,d)\,\left(A\,b^2\,(m+2)+C\,\left(a^2+b^2\,(m+1)\right)\right)\right)\,\text{Sin}[e+f\,x]-$$
$$b\,C\,d\,(m+1)\,\left(a^2-b^2\right)\,\text{Sin}[e+f\,x]^2\Big)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(b*c-a*d)*(A*b^2-a*b*B+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b^2*f*(m+1)*(a^2-b^2)) -
  1/(b^2*(m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*
    Simp[b*(m+1)*((b*B-a*C)*(b*c-a*d)-A*b*(a*c-b*d))+
      (b*B*(a^2*d+b^2*d*(m+1)-a*b*c*(m+2))+(b*c-a*d)*(A*b^2*(m+2)+C*(a^2+b^2*(m+1))))*Sin[e+f*x]-
     b*C*d*(m+1)*(a^2-b^2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(b*c-a*d)*(A*b^2+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b^2*f*(m+1)*(a^2-b^2)) +
  1/(b^2*(m+1)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*
    Simp[b*(m+1)*(a*C*(b*c-a*d)+A*b*(a*c-b*d))-
      ((b*c-a*d)*(A*b^2*(m+2)+C*(a^2+b^2*(m+1))))*Sin[e+f*x]+
      b*C*d*(m+1)*(a^2-b^2)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m\not<-1$

Derivation: Algebraic expansion, nondegenerate sine recurrence 1b with $c\to 0,\ d\to 1,\ A\to a\,c,\ B\to b\,c+a\,d,\ C\to b\,d,\ m\to m+1,\ n\to 0,\ p\to 0$ and algebraic simplification

Basis: $A+B\,z+C\,z^2 == \frac{C\,(a+b\,z)^2}{b^2}+\frac{A\,b^2-a^2\,C+b\,(b\,B-2\,a\,C)\,z}{b^2}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,m\not<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{C}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^{m+2}\,(c+d\,\text{Sin}[e+f\,x])\,dx+\frac{1}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])\,\left(A\,b^2-a^2\,C+b\,(b\,B-2\,a\,C)\,\text{Sin}[e+f\,x]\right)\,dx\,\longrightarrow$$

$$-\frac{C\,d\,\text{Cos}[e+f\,x]\,\text{Sin}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^{m+1}}{b\,f\,(m+3)}+\frac{1}{b\,(m+3)}\int (a+b\,\text{Sin}[e+f\,x])^m\cdot$$
$$\left(a\,C\,d+A\,b\,c\,(m+3)+b\,(B\,c\,(m+3)+d\,(C\,(m+2)+A\,(m+3)))\,\text{Sin}[e+f\,x]-(2\,a\,C\,d-b\,(c\,C+B\,d)\,(m+3))\,\text{Sin}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*d*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+3)) +
  1/(b*(m+3))*Int[(a+b*Sin[e+f*x])^m*
     Simp[a*C*d+A*b*c*(m+3)+b*(B*c*(m+3)+d*(C*(m+2)+A*(m+3)))*Sin[e+f*x]-(2*a*C*d-b*(c*C+B*d)*(m+3))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_+d_.*sin[e_.+f_.*x_])*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*d*Cos[e+f*x]*Sin[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(m+3)) +
  1/(b*(m+3))*Int[(a+b*Sin[e+f*x])^m*
    Simp[a*C*d+A*b*c*(m+3)+b*d*(C*(m+2)+A*(m+3))*Sin[e+f*x]-(2*a*C*d-b*c*C*(m+3))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && Not[LtQ[m,-1]]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m<-\frac{1}{2}$

Derivation: Algebraic expansion, singly degenerate sine recurrence 2b with $A\to 1,\ B\to 0,\ p\to 0$ and algebraic simplification

Basis: If $a^2-b^2==0$, then $A+B\,z+C\,z^2==\frac{a\,A-b\,B+a\,C}{a}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m<-\frac{1}{2}$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{a\,A-b\,B+a\,C}{a}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx+\frac{1}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\,(b\,B-a\,C+b\,C\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(a\,A-b\,B+a\,C)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{2\,b\,c\,f\,(2\,m+1)}-$$
$$\frac{1}{2\,b\,c\,d\,(2\,m+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\cdot$$
$$\left(A\left(c^2\,(m+1)+d^2\,(2\,m+n+2)\right)-B\,c\,d\,(m-n-1)-C\left(c^2\,m-d^2\,(n+1)\right)+d\,((A\,c+B\,d)\,(m+n+2)-c\,C\,(3\,m-n))\,\text{Sin}[e+f\,x]\right)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (a*A-b*B+a*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(2*b*c*f*(2*m+1)) -
  1/(2*b*c*d*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[A*(c^2*(m+1)+d^2*(2*m+n+2))-B*c*d*(m-n-1)-C*(c^2*m-d^2*(n+1))+d*((A*c+B*d)*(m+n+2)-c*C*(3*m-n))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (LtQ[m,-1/2] || EqQ[m+n+2,0] && NeQ[2*m+1,0])
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (a*A+a*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(2*b*c*f*(2*m+1)) -
  1/(2*b*c*d*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[A*(c^2*(m+1)+d^2*(2*m+n+2))-C*(c^2*m-d^2*(n+1))+d*(A*c*(m+n+2)-c*C*(3*m-n))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && (LtQ[m,-1/2] || EqQ[m+n+2,0] && NeQ[2*m+1,0])
```

2. $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2)\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}$

1: $\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2)}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$ when $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}$

Derivation: Algebraic expansion and doubly degenerate sine recurrence 2b with $n\to-\frac{1}{2}$, $p\to 0$

Basis: $A+B\,z+C\,z^2=\frac{C\,(e+f\,z+g\,z^2)}{g}-\frac{C\,e-A\,g+(C\,f-B\,g)\,z}{g}$

Rule: If $b\,c+a\,d=0\,\wedge\,a^2-b^2=0\,\wedge\,m\not<-\frac{1}{2}$, then

$$\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2)}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx\,\longrightarrow$$

$$-\frac{2\,C\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}}{b\,f\,(2\,m+3)\,\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}+\int \frac{(a+b\,\mathrm{Sin}[e+f\,x])^m\,(A+C+B\,\mathrm{Sin}[e+f\,x])}{\sqrt{c+d\,\mathrm{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/Sqrt[c_.+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2*C*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(2*m+3)*Sqrt[c+d*Sin[e+f*x]]) +
  Int[(a+b*Sin[e+f*x])^m*Simp[A+C+B*Sin[e+f*x],x]/Sqrt[c+d*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(A_.+C_.*sin[e_.+f_.*x_]^2)/Sqrt[c_.+d_.*sin[e_.+f_.*x_]],x_Symbol] :=
  -2*C*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)/(b*f*(2*m+3)*Sqrt[c+d*Sin[e+f*x]]) +
  (A+C)*Int[(a+b*Sin[e+f*x])^m/Sqrt[c+d*Sin[e+f*x]],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m\nless-\frac{1}{2}\,\land\,m+n+2\neq0$

Derivation: Nondegenerate sine recurrence 1b with $p\to0$ and $a^2-b^2==0$

Derivation: Algebraic expansion and singly degenerate sine recurrence 2c with $A\to c,\ B\to d,\ n\to n+1,\ p\to0$

Basis: $A+B\,z+C\,z^2==\frac{C\,(c+d\,z)^2}{d^2}+\frac{A\,d^2-c^2\,C-d\,(2\,c\,C-B\,d)\,z}{d^2}$

Rule: If $b\,c+a\,d==0\,\land\,a^2-b^2==0\,\land\,m\nless-\frac{1}{2}\,\land\,m+n+2\neq0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow$$

$$\frac{C}{d^2}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+2}\,dx+\frac{1}{d^2}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A\,d^2-c^2\,C-d\,(2\,c\,C-B\,d)\,\mathrm{Sin}[e+f\,x]\right)\,dx\,\longrightarrow$$

$$-\frac{C\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n+2)}+$$

$$\frac{1}{b\,d\,(m+n+2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,(A\,b\,d\,(m+n+2)+C\,(a\,c\,m+b\,d\,(n+1))+(b\,B\,d\,(m+n+2)-b\,c\,C\,(2\,m+1))\,\mathrm{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(b*d*(m+n+2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*
    Simp[A*b*d*(m+n+2)+C*(a*c*m+b*d*(n+1))+(b*B*d*(m+n+2)-b*c*C*(2*m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && NeQ[m+n+2,0]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(b*d*(m+n+2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*
    Simp[A*b*d*(m+n+2)+C*(a*c*m+b*d*(n+1))-b*c*C*(2*m+1)*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]] && NeQ[m+n+2,0]
```

3. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m<-\frac{1}{2}$

Derivation: Algebraic expansion, singly degenerate sine recurrence 2b with $A\to 1,\ B\to 0,\ p\to 0$ and algebraic simplification

Basis: If $a^2-b^2==0$, then $A+B\,z+C\,z^2==\frac{a\,A-b\,B+a\,C}{a}+\frac{(a+b\,z)\,(b\,B-a\,C+b\,C\,z)}{b^2}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,m<-\frac{1}{2}$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$\frac{a\,A-b\,B+a\,C}{a}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx+\frac{1}{b^2}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\,(b\,B-a\,C+b\,C\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{(a\,A-b\,B+a\,C)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{f\,(b\,c-a\,d)\,(2\,m+1)}+$$
$$\frac{1}{b\,(b\,c-a\,d)\,(2\,m+1)}\int (a+b\,\text{Sin}[e+f\,x])^{m+1}\,(c+d\,\text{Sin}[e+f\,x])^n\cdot$$
$$(A\,(a\,c\,(m+1)-b\,d\,(2\,m+n+2))+B\,(b\,c\,m+a\,d\,(n+1))-C\,(a\,c\,m+b\,d\,(n+1))+(d\,(a\,A-b\,B)\,(m+n+2)+C\,(b\,c\,(2\,m+1)-a\,d\,(m-n-1)))\,\text{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (a*A-b*B+a*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(f*(b*c-a*d)*(2*m+1)) +
  1/(b*(b*c-a*d)*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[A*(a*c*(m+1)-b*d*(2*m+n+2))+B*(b*c*m+a*d*(n+1))-C*(a*c*m+b*d*(n+1))+
      (d*(a*A-b*B)*(m+n+2)+C*(b*c*(2*m+1)-a*d*(m-n-1)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1/2]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  a*(A+C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(f*(b*c-a*d)*(2*m+1)) +
  1/(b*(b*c-a*d)*(2*m+1))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[A*(a*c*(m+1)-b*d*(2*m+n+2))-C*(a*c*m+b*d*(n+1))+
      (a*A*d*(m+n+2)+C*(b*c*(2*m+1)-a*d*(m-n-1)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1/2]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\nless -\frac{1}{2}$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\nless -\frac{1}{2}\,\land\,(n<-1\,\lor\,m+n+2==0)$

Derivation: Algebraic expansion and singly degenerate sine recurrence 1c with $A\to 1,\ B\to 0,\ p\to 0$

Basis: $A+B\,z+C\,z^2==\frac{c^2\,C-B\,c\,d+A\,d^2}{d^2}-\frac{(c+d\,z)\,(c\,C-B\,d-C\,d\,z)}{d^2}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\nless -\frac{1}{2}\,\land\,(n<-1\,\lor\,m+n+2==0)$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$\frac{c^2\,C-B\,c\,d+A\,d^2}{d^2}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,dx-\frac{1}{d^2}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\,(c\,C-B\,d-C\,d\,\text{Sin}[e+f\,x])\,dx\,\longrightarrow$$

$$-\frac{(c^2\,C-B\,c\,d+A\,d^2)\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,(c^2-d^2)}+$$
$$\frac{1}{b\,d\,(n+1)\,(c^2-d^2)}\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\cdot$$
$$(A\,d\,(a\,d\,m+b\,c\,(n+1))+(c\,C-B\,d)\,(a\,c\,m+b\,d\,(n+1))+b\,(d\,(B\,c-A\,d)\,(m+n+2)-C\,(c^2\,(m+1)+d^2\,(n+1)))\,\text{Sin}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(c^2*C-B*c*d+A*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(b*d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)*
    Simp[A*d*(a*d*m+b*c*(n+1))+(c*C-B*d)*(a*c*m+b*d*(n+1))+b*(d*(B*c-A*d)*(m+n+2)-C*(c^2*(m+1)+d^2*(n+1)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[LtQ[m,-1/2]] && (LtQ[n,-1] || EqQ[m+n+2,0])
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(c^2*C+A*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(b*d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)*
    Simp[A*d*(a*d*m+b*c*(n+1))+c*C*(a*c*m+b*d*(n+1))-b*(A*d^2*(m+n+2)+C*(c^2*(m+1)+d^2*(n+1)))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[LtQ[m,-1/2]] && (LtQ[n,-1] || EqQ[m+n+2,0])
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\not<-\frac{1}{2}\,\land\,m+n+2\neq 0$

Derivation: Nondegenerate sine recurrence 1b with $p\to 0$ and $a^2-b^2==0$

Derivation: Algebraic expansion and singly degenerate sine recurrence 2c with $A\to c,\ B\to d,\ n\to n+1,\ p\to 0$

Basis: $A+B\,z+C\,z^2==\frac{C\,(c+d\,z)^2}{d^2}+\frac{A\,d^2-c^2\,C-d\,(2\,c\,C-B\,d)\,z}{d^2}$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2==0\,\land\,c^2-d^2\neq 0\,\land\,m\not<-\frac{1}{2}\,\land\,m+n+2\neq 0$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)dx\,\longrightarrow$$

$$\frac{C}{d^2}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+2}\,dx+\frac{1}{d^2}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A\,d^2-c^2\,C-d\,(2\,c\,C-B\,d)\,\mathrm{Sin}[e+f\,x]\right)dx\,\longrightarrow$$

$$-\frac{C\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n+2)}+$$

$$\frac{1}{b\,d\,(m+n+2)}\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A\,b\,d\,(m+n+2)+C\,(a\,c\,m+b\,d\,(n+1))+(C\,(a\,d\,m-b\,c\,(m+1))+b\,B\,d\,(m+n+2))\,\mathrm{Sin}[e+f\,x]\right)dx$$

Program code:

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(b*d*(m+n+2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*
    Simp[A*b*d*(m+n+2)+C*(a*c*m+b*d*(n+1))+(C*(a*d*m-b*c*(m+1))+b*B*d*(m+n+2))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[LtQ[m,-1/2]] && NeQ[m+n+2,0]
```

```
Int[(a_+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(b*d*(m+n+2))*Int[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*
    Simp[A*b*d*(m+n+2)+C*(a*c*m+b*d*(n+1))+C*(a*d*m-b*c*(m+1))*Sin[e+f*x],x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && NeQ[b*c-a*d,0] && EqQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && Not[LtQ[m,-1/2]] && NeQ[m+n+2,0]
```

4. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0$

1. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m>0$

1: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m>0\,\land\,n<-1$

Derivation: Nondegenerate sine recurrence 1a with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m>0\,\land\,n<-1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx\,\longrightarrow$$

$$-\frac{\left(c^2\,C-B\,c\,d+A\,d^2\right)\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(n+1)\,\left(c^2-d^2\right)}+$$

$$\frac{1}{d\,(n+1)\,\left(c^2-d^2\right)}\int (a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^{n+1}\cdot$$

$$\Big(A\,d\,(b\,d\,m+a\,c\,(n+1))+(c\,C-B\,d)\,(b\,c\,m+a\,d\,(n+1))\,-$$
$$\left(d\,(A\,(a\,d\,(n+2)-b\,c\,(n+1))+B\,(b\,d\,(n+1)-a\,c\,(n+2)))-C\left(b\,c\,d\,(n+1)-a\left(c^2+d^2\,(n+1)\right)\right)\right)\text{Sin}[e+f\,x]+$$
$$b\left(d\,(B\,c-A\,d)\,(m+n+2)-C\left(c^2\,(m+1)+d^2\,(n+1)\right)\right)\text{Sin}[e+f\,x]^2\Big)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(c^2*C-B*c*d+A*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[A*d*(b*d*m+a*c*(n+1))+(c*C-B*d)*(b*c*m+a*d*(n+1)) -
      (d*(A*(a*d*(n+2)-b*c*(n+1))+B*(b*d*(n+1)-a*c*(n+2)))-C*(b*c*d*(n+1)-a*(c^2+d^2*(n+1))))*Sin[e+f*x] +
       b*(d*(B*c-A*d)*(m+n+2)-C*(c^2*(m+1)+d^2*(n+1)))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,0] && LtQ[n,-1]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(c^2*C+A*d^2)*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(n+1)*(c^2-d^2)) +
  1/(d*(n+1)*(c^2-d^2))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^(n+1)*
    Simp[A*d*(b*d*m+a*c*(n+1))+c*C*(b*c*m+a*d*(n+1)) -
      (A*d*(a*d*(n+2)-b*c*(n+1))-C*(b*c*d*(n+1)-a*(c^2+d^2*(n+1))))*Sin[e+f*x] -
      b*(A*d^2*(m+n+2)+C*(c^2*(m+1)+d^2*(n+1)))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,0] && LtQ[n,-1]
```

2: $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m>0\,\land\,n\nless -1$

Derivation: Nondegenerate sine recurrence 1b with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m>0\,\land\,n\nless -1$, then

$$\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx\longrightarrow$$
$$-\frac{C\,\text{Cos}[e+f\,x]\,(a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^{n+1}}{d\,f\,(m+n+2)}+$$
$$\frac{1}{d\,(m+n+2)}\int (a+b\,\text{Sin}[e+f\,x])^{m-1}\,(c+d\,\text{Sin}[e+f\,x])^n\cdot$$
$$\left(a\,A\,d\,(m+n+2)+C\,(b\,c\,m+a\,d\,(n+1))+(d\,(A\,b+a\,B)\,(m+n+2)-C\,(a\,c-b\,d\,(m+n+1)))\,\text{Sin}[e+f\,x]+(C\,(a\,d\,m-b\,c\,(m+1))+b\,B\,d\,(m+n+2))\,\text{Sin}[e+f\,x]^2\right)dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(d*(m+n+2))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n*
     Simp[a*A*d*(m+n+2)+C*(b*c*m+a*d*(n+1))+
      (d*(A*b+a*B)*(m+n+2)-C*(a*c-b*d*(m+n+1)))*Sin[e+f*x]+
      (C*(a*d*m-b*c*(m+1))+b*B*d*(m+n+2))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,0] &&
  Not[IGtQ[n,0] && (Not[IntegerQ[m]] || EqQ[a,0] && NeQ[c,0])]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_.*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cos[e+f*x]*(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^(n+1)/(d*f*(m+n+2)) +
  1/(d*(m+n+2))*Int[(a+b*Sin[e+f*x])^(m-1)*(c+d*Sin[e+f*x])^n*
    Simp[a*A*d*(m+n+2)+C*(b*c*m+a*d*(n+1))+(A*b*d*(m+n+2)-C*(a*c-b*d*(m+n+1)))*Sin[e+f*x]+C*(a*d*m-b*c*(m+1))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && GtQ[m,0] &&
  Not[IGtQ[n,0] && (Not[IntegerQ[m]] || EqQ[a,0] && NeQ[c,0])]
```

2. $\int (a+b\,\text{Sin}[e+f\,x])^m\,(c+d\,\text{Sin}[e+f\,x])^n\,\left(A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2\right)dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0\,\wedge\,m<-1$

1. $\int \frac{A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{c+d\,\text{Sin}[e+f\,x]}}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,a^2-b^2\neq 0\,\wedge\,c^2-d^2\neq 0$

1: $\int \frac{A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B\,z+C\,z^2}{(a+b\,z)^{3/2}\sqrt{d\,z}}==\frac{C\,\sqrt{d\,z}}{b\,d\,\sqrt{a+b\,z}}+\frac{A\,b+(b\,B-a\,C)\,z}{b\,(a+b\,z)^{3/2}\sqrt{d\,z}}$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{A+B\,\text{Sin}[e+f\,x]+C\,\text{Sin}[e+f\,x]^2}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx\to\frac{C}{b\,d}\int\frac{\sqrt{d\,\text{Sin}[e+f\,x]}}{\sqrt{a+b\,\text{Sin}[e+f\,x]}}\,dx+\frac{1}{b}\int\frac{A\,b+(b\,B-a\,C)\,\text{Sin}[e+f\,x]}{(a+b\,\text{Sin}[e+f\,x])^{3/2}\sqrt{d\,\text{Sin}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  C/(b*d)*Int[Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] +
  1/b*Int[(A*b+(b*B-a*C)*Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f,A,B,C},x] && NeQ[a^2-b^2,0]
```

```
Int[(A_.+C_.*sin[e_.+f_.*x_]^2)/((a_+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  C/(b*d)*Int[Sqrt[d*Sin[e+f*x]]/Sqrt[a+b*Sin[e+f*x]],x] +
  1/b*Int[(A*b-a*C*Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,d,e,f,A,C},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{(a + b \operatorname{Sin}[e + f x])^{3/2} \sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{(a+b z)^{3/2}} == \frac{C \sqrt{a+b z}}{b^2} + \frac{A b^2-a^2 C+b (b B-2 a C) z}{b^2 (a+b z)^{3/2}}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{(a + b \operatorname{Sin}[e + f x])^{3/2} \sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx \to \frac{C}{b^2} \int \frac{\sqrt{a + b \operatorname{Sin}[e + f x]}}{\sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx + \frac{1}{b^2} \int \frac{A b^2 - a^2 C + b (b B - 2 a C) \operatorname{Sin}[e + f x]}{(a + b \operatorname{Sin}[e + f x])^{3/2} \sqrt{c + d \operatorname{Sin}[e + f x]}} \, dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/((a_.+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  C/b^2*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  1/b^2*Int[(A*b^2-a^2*C+b*(b*B-2*a*C)*Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

```
Int[(A_.+C_.*sin[e_.+f_.*x_]^2)/((a_.+b_.*sin[e_.+f_.*x_])^(3/2)*Sqrt[c_.+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  C/b^2*Int[Sqrt[a+b*Sin[e+f*x]]/Sqrt[c+d*Sin[e+f*x]],x] +
  1/b^2*Int[(A*b^2-a^2*C-2*a*b*C*Sin[e+f*x])/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

2: $\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx$ when $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1$

Derivation: Nondegenerate sine recurrence 1c with $p\to 0$

Rule: If $b\,c-a\,d\neq 0\,\land\,a^2-b^2\neq 0\,\land\,c^2-d^2\neq 0\,\land\,m<-1$, then

$$\int (a+b\,\mathrm{Sin}[e+f\,x])^m\,(c+d\,\mathrm{Sin}[e+f\,x])^n\,\left(A+B\,\mathrm{Sin}[e+f\,x]+C\,\mathrm{Sin}[e+f\,x]^2\right)\,dx\,\longrightarrow$$
$$-\frac{\left(A\,b^2-a\,b\,B+a^2\,C\right)\,\mathrm{Cos}[e+f\,x]\,(a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^{n+1}}{f\,(m+1)\,(b\,c-a\,d)\,\left(a^2-b^2\right)}+$$
$$\frac{1}{(m+1)\,(b\,c-a\,d)\,\left(a^2-b^2\right)}\int (a+b\,\mathrm{Sin}[e+f\,x])^{m+1}\,(c+d\,\mathrm{Sin}[e+f\,x])^n\cdot$$
$$\left((m+1)\,(b\,c-a\,d)\,(a\,A-b\,B+a\,C)+d\,\left(A\,b^2-a\,b\,B+a^2\,C\right)\,(m+n+2)-\right.$$
$$\left(c\,\left(A\,b^2-a\,b\,B+a^2\,C\right)+(m+1)\,(b\,c-a\,d)\,(A\,b-a\,B+b\,C)\right)\,\mathrm{Sin}[e+f\,x]-$$
$$\left.d\,\left(A\,b^2-a\,b\,B+a^2\,C\right)\,(m+n+3)\,\mathrm{Sin}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2-a*b*B+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2-b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[(m+1)*(b*c-a*d)*(a*A-b*B+a*C)+d*(A*b^2-a*b*B+a^2*C)*(m+n+2) -
      (c*(A*b^2-a*b*B+a^2*C)+(m+1)*(b*c-a*d)*(A*b-a*B+b*C))*Sin[e+f*x] -
      d*(A*b^2-a*b*B+a^2*C)*(m+n+3)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] &&
  (EqQ[a,0] && IntegerQ[m] && Not[IntegerQ[n]] || Not[IntegerQ[2*n] && LtQ[n,-1] && (IntegerQ[n] && Not[IntegerQ[m]] || EqQ[a,0])])
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  -(A*b^2+a^2*C)*Cos[e+f*x]*(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^(n+1)/(f*(m+1)*(b*c-a*d)*(a^2-b^2)) +
  1/((m+1)*(b*c-a*d)*(a^2-b^2))*Int[(a+b*Sin[e+f*x])^(m+1)*(c+d*Sin[e+f*x])^n*
    Simp[a*(m+1)*(b*c-a*d)*(A+C)+d*(A*b^2+a^2*C)*(m+n+2) -
      (c*(A*b^2+a^2*C)+b*(m+1)*(b*c-a*d)*(A+C))*Sin[e+f*x] -
      d*(A*b^2+a^2*C)*(m+n+3)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0] && LtQ[m,-1] &&
  (EqQ[a,0] && IntegerQ[m] && Not[IntegerQ[n]] || Not[IntegerQ[2*n] && LtQ[n,-1] && (IntegerQ[n] && Not[IntegerQ[m]] || EqQ[a,0])])
```

3: $\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{(a + b \operatorname{Sin}[e + f x]) (c + d \operatorname{Sin}[e + f x])} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{(a+b z) (c+d z)} == \frac{C}{b d} + \frac{A b^2-a b B+a^2 C}{b (b c-a d) (a+b z)} - \frac{c^2 C-B c d+A d^2}{d (b c-a d) (c+d z)}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{(a + b \operatorname{Sin}[e + f x]) (c + d \operatorname{Sin}[e + f x])} \, dx \longrightarrow$$
$$\frac{C x}{b d} + \frac{A b^2 - a b B + a^2 C}{b (b c - a d)} \int \frac{1}{a + b \operatorname{Sin}[e + f x]} \, dx - \frac{c^2 C - B c d + A d^2}{d (b c - a d)} \int \frac{1}{c + d \operatorname{Sin}[e + f x]} \, dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/((a_+b_.*sin[e_.+f_.*x_])*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  C*x/(b*d) +
  (A*b^2-a*b*B+a^2*C)/(b*(b*c-a*d))*Int[1/(a+b*Sin[e+f*x]),x] -
  (c^2*C-B*c*d+A*d^2)/(d*(b*c-a*d))*Int[1/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

```
Int[(A_.+C_.*sin[e_.+f_.*x_]^2)/((a_+b_.*sin[e_.+f_.*x_])*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  C*x/(b*d) +
  (A*b^2+a^2*C)/(b*(b*c-a*d))*Int[1/(a+b*Sin[e+f*x]),x] -
  (c^2*C+A*d^2)/(d*(b*c-a*d))*Int[1/(c+d*Sin[e+f*x]),x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

4: $\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{\sqrt{a + b \operatorname{Sin}[e + f x]}\, (c + d \operatorname{Sin}[e + f x])} \, dx$ when $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{\sqrt{a+b z}\, (c+d z)} == \frac{C \sqrt{a+b z}}{b d} - \frac{a c C-A b d+(b c C-b B d+a C d)\, z}{b d \sqrt{a+b z}\, (c+d z)}$

Rule: If $b c - a d \neq 0 \land a^2 - b^2 \neq 0 \land c^2 - d^2 \neq 0$, then

$$\int \frac{A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2}{\sqrt{a + b \operatorname{Sin}[e + f x]}\, (c + d \operatorname{Sin}[e + f x])} \, dx \longrightarrow$$
$$\frac{C}{b d} \int \sqrt{a + b \operatorname{Sin}[e + f x]} \, dx - \frac{1}{b d} \int \frac{a c C - A b d + (b c C - b B d + a C d) \operatorname{Sin}[e + f x]}{\sqrt{a + b \operatorname{Sin}[e + f x]}\, (c + d \operatorname{Sin}[e + f x])} \, dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/(Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  C/(b*d)*Int[Sqrt[a+b*Sin[e+f*x]],x] -
  1/(b*d)*Int[Simp[a*c*C-A*b*d+(b*c*C-b*B*d+a*C*d)*Sin[e+f*x],x]/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

```
Int[(A_.+C_.*sin[e_.+f_.*x_]^2)/(Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*(c_.+d_.*sin[e_.+f_.*x_])),x_Symbol] :=
  C/(b*d)*Int[Sqrt[a+b*Sin[e+f*x]],x] -
  1/(b*d)*Int[Simp[a*c*C-A*b*d+(b*c*C+a*C*d)*Sin[e+f*x],x]/(Sqrt[a+b*Sin[e+f*x]]*(c+d*Sin[e+f*x])),x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

5: $\int \frac{A + B \sin[e + f x] + C \sin[e + f x]^2}{\sqrt{a + b \sin[e + f x]} \sqrt{c + d \sin[e + f x]}} \, dx$ when $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge c^2 - d^2 \neq 0$

Derivation: Nondegenerate sine recurrence 1b with $m \to -\frac{1}{2}$, $n \to -\frac{1}{2}$, $p \to 0$

Note: If one of the square root factors does not have a constant term, it is better to raise that factor to the 3/2 power.

Rule: If $b c - a d \neq 0 \wedge a^2 - b^2 \neq 0 \wedge c^2 - d^2 \neq 0$, then

$$\int \frac{A + B \sin[e + f x] + C \sin[e + f x]^2}{\sqrt{a + b \sin[e + f x]} \sqrt{c + d \sin[e + f x]}} \, dx \longrightarrow$$

$$-\frac{C \cos[e + f x] \sqrt{c + d \sin[e + f x]}}{d f \sqrt{a + b \sin[e + f x]}} +$$

$$\frac{1}{2 d} \int \left(\left(2 a A d - C (b c - a d) - 2 (a c C - d (A b + a B)) \sin[e + f x] + (2 b B d - C (b c + a d)) \sin[e + f x]^2\right) \Big/ \left((a + b \sin[e + f x])^{3/2} \sqrt{c + d \sin[e + f x]} \right) \right) dx$$

Program code:

```
Int[(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/(Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -C*Cos[e+f*x]*Sqrt[c+d*Sin[e+f*x]]/(d*f*Sqrt[a+b*Sin[e+f*x]]) +
  1/(2*d)*Int[1/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]])*
    Simp[2*a*A*d-C*(b*c-a*d)-2*(a*c*C-d*(A*b+a*B))*Sin[e+f*x]+(2*b*B*d-C*(b*c+a*d))*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,B,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

```
Int[(A_.+C_.*sin[e_.+f_.*x_]^2)/(Sqrt[a_.+b_.*sin[e_.+f_.*x_]]*Sqrt[c_+d_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  -C*Cos[e+f*x]*Sqrt[c+d*Sin[e+f*x]]/(d*f*Sqrt[a+b*Sin[e+f*x]]) +
  1/(2*d)*Int[1/((a+b*Sin[e+f*x])^(3/2)*Sqrt[c+d*Sin[e+f*x]])*
    Simp[2*a*A*d-C*(b*c-a*d)-2*(a*c*C-A*b*d)*Sin[e+f*x]-C*(b*c+a*d)*Sin[e+f*x]^2,x],x] /;
FreeQ[{a,b,c,d,e,f,A,C},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

6: $\int \frac{(d \sin[e+f x])^n \left(A+B \sin[e+f x]+C \sin[e+f x]^2\right)}{a+b \sin[e+f x]} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{a+b z} == \frac{b B-a C}{b^2}+\frac{C z}{b}+\frac{A b^2-a b B+a^2 C}{b^2 (a+b z)}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{(d \sin[e+f x])^n \left(A+B \sin[e+f x]+C \sin[e+f x]^2\right)}{a+b \sin[e+f x]} \, dx \rightarrow$$

$$\frac{b B-a C}{b^2} \int (d \sin[e+f x])^n \, dx+\frac{C}{b d} \int (d \sin[e+f x])^{n+1} \, dx+\frac{A b^2-a b B+a^2 C}{b^2} \int \frac{(d \sin[e+f x])^n}{a+b \sin[e+f x]} \, dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2)/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  (b*B-a*C)/b^2*Int[(d*Sin[e+f*x])^n,x] +
  C/(b*d)*Int[(d*Sin[e+f*x])^(n+1),x] +
  (A*b^2-a*b*B+a^2*C)/b^2*Int[(d*Sin[e+f*x])^n/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,B,C,n},x] && NeQ[a^2-b^2,0]
```

```
Int[(d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2)/(a_+b_.*sin[e_.+f_.*x_]),x_Symbol] :=
  -a*C/b^2*Int[(d*Sin[e+f*x])^n,x] +
  C/(b*d)*Int[(d*Sin[e+f*x])^(n+1),x] +
  (A*b^2+a^2*C)/b^2*Int[(d*Sin[e+f*x])^n/(a+b*Sin[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,C,n},x] && NeQ[a^2-b^2,0]
```

U: $\int (a+b \sin[e+f x])^m (c+d \sin[e+f x])^n \left(A+B \sin[e+f x]+C \sin[e+f x]^2\right) dx$ when $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$

Rule: If $b c-a d \neq 0 \land a^2-b^2 \neq 0 \land c^2-d^2 \neq 0$, then

$$\int (a+b \sin[e+f x])^m (c+d \sin[e+f x])^n \left(A+B \sin[e+f x]+C \sin[e+f x]^2\right) dx \rightarrow$$

$$\int (a + b \operatorname{Sin}[e + f x])^m (c + d \operatorname{Sin}[e + f x])^n (A + B \operatorname{Sin}[e + f x] + C \operatorname{Sin}[e + f x]^2) \, dx$$

Program code:

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  Unintegrable[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*(A+B*Sin[e+f*x]+C*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,B,C,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

```
Int[(a_.+b_.*sin[e_.+f_.*x_])^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  Unintegrable[(a+b*Sin[e+f*x])^m*(c+d*Sin[e+f*x])^n*(A+C*Sin[e+f*x]^2),x] /;
FreeQ[{a,b,c,d,e,f,A,C,m,n},x] && NeQ[b*c-a*d,0] && NeQ[a^2-b^2,0] && NeQ[c^2-d^2,0]
```

Rules for integrands of the form $(b \sin[e+f x]^p)^m (c+d \sin[e+f x])^n (A+B \sin[e+f x]+C \sin[e+f x]^2)$

1: $\int (b \sin[e+f x]^p)^m (c+d \sin[e+f x])^n (A+B \sin[e+f x]+C \sin[e+f x]^2)\, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b \sin[e+f x]^p)^m}{(b \sin[e+f x])^{m p}} == 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (b \sin[e+f x]^p)^m (c+d \sin[e+f x])^n (A+B \sin[e+f x]+C \sin[e+f x]^2)\, dx \rightarrow$$
$$\frac{(b \sin[e+f x]^p)^m}{(b \sin[e+f x])^{m p}} \int (b \sin[e+f x])^{m p} (c+d \sin[e+f x])^n (A+B \sin[e+f x]+C \sin[e+f x]^2)\, dx$$

Program code:

```
Int[(b_.*sin[e_.+f_.*x_]^p_)^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+B_.*sin[e_.+f_.*x_]+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Sin[e+f*x]^p)^m/(b*Sin[e+f*x])^(m*p)*Int[(b*Sin[e+f*x])^(m*p)*(c+d*Sin[e+f*x])^n*(A+B*Sin[e+f*x]+C*Sin[e+f*x]^2),x] /;
FreeQ[{b,c,d,e,f,A,B,C,m,n,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*cos[e_.+f_.*x_]^p_)^m_*(c_.+d_.*cos[e_.+f_.*x_])^n_.*(A_.+B_.*cos[e_.+f_.*x_]+C_.*cos[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Cos[e+f*x]^p)^m/(b*Cos[e+f*x])^(m*p)*Int[(b*Cos[e+f*x])^(m*p)*(c+d*Cos[e+f*x])^n*(A+B*Cos[e+f*x]+C*Cos[e+f*x]^2),x] /;
FreeQ[{b,c,d,e,f,A,B,C,m,n,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*sin[e_.+f_.*x_]^p_)^m_*(c_.+d_.*sin[e_.+f_.*x_])^n_.*(A_.+C_.*sin[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Sin[e+f*x]^p)^m/(b*Sin[e+f*x])^(m*p)*Int[(b*Sin[e+f*x])^(m*p)*(c+d*Sin[e+f*x])^n*(A+C*Sin[e+f*x]^2),x] /;
FreeQ[{b,c,d,e,f,A,C,m,n,p},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*cos[e_.+f_.*x_]^p_)^m_*(c_.+d_.*cos[e_.+f_.*x_])^n_.*(A_.+C_.*cos[e_.+f_.*x_]^2),x_Symbol] :=
  (b*Cos[e+f*x]^p)^m/(b*Cos[e+f*x])^(m*p)*Int[(b*Cos[e+f*x])^(m*p)*(c+d*Cos[e+f*x])^n*(A+C*Cos[e+f*x]^2),x] /;
FreeQ[{b,c,d,e,f,A,C,m,n,p},x] && Not[IntegerQ[m]]
```